//
// This program is free software: you can redistribute it and/or modify
// it under the terms of the GNU Lesser General Public License as published by
// the Free Software Foundation, either version 3 of the License, or
// (at your option) any later version.
// 
// This program is distributed in the hope that it will be useful,
// but WITHOUT ANY WARRANTY; without even the implied warranty of
// MERCHANTABILITY or FITNESS FOR A PARTICULAR PURPOSE.  See the
// GNU Lesser General Public License for more details.
// 
// You should have received a copy of the GNU Lesser General Public License
// along with this program.  If not, see http://www.gnu.org/licenses/.
// 

package peermaliziosi.simulations;

import peermaliziosi.Peer;
import peermaliziosi.Cloud;

network BitTorrent
{
    parameters:
        int numPeers = default(55);
        int numSeeders= default(1); // numero di seeder iniziali del torrent (dovrebbe sempre essere minore al numPeers)
        // pezzi da 256KiB;blocchi da 16KiB => 16 blocchi per pezzo
		// file da 100MiB => 400 pezzi
        int numPieces = default(400); // numero pezzi
        int numBlocks = default(16);  // numero di blocchi per pezzo
        int remPiecesNeedSeeder= default(100); // indica al tracker che deve restituire 1 seeder ogni 100 pezzi che mancano al peer
         // dimensione del blocco da trasferire in byte
        int blockLength @unit(byte) = default(16KiB);

        @display("bgb=668,534");
    types:
        // ricorda che puo' esserci al piu' un canale "ned.DatarateChannel" nel path
        // il canale in uscita simula tutti i tempi di trasmissione dei pacchetti
        channel MyDatarate extends ned.DatarateChannel {
            datarate = default(4Mbps);
            // delay peer-peer: 30ms (round-trip-time = 60ms) 
            delay = default(15ms);
        }
        // il canale di ingresso e' ideale
        channel MyIdeal extends ned.IdealChannel {}
    submodules:
        peer[numPeers]: Peer {
            @display("p=300,120;is=s");
        }
        cloud: Cloud {
            @display("p=300,300;is=vl");
            gates: io[numPeers];
        }
    connections:
        // il gate i-esimo e' collegato al peer i-esimo
        // TODO in realta' questo modo e' SBAGLIATO. Esempio: Peer-A, Peer-B e Peer-C hanno tutti lo stesso
        // datarate. Peer-A e Peer-B inviano entrambi a Peer-C
        // un messaggio con durata di trasmissione 1s. Entrambi i msg arrivano nella nuovola a distanza di 0.5s
        // e la nuvola gli inoltra istantaneamente a Peer-C che li vede arrivare a distanza di 0.5s, quando
        // dovrebbe invece essere a distanza di 1s
        for i=0..numPeers-1 {
            cloud.io[i] <--> MyDatarate <--> peer[i].io;
        }
}
